

# FAQ WinDev

Date de publication : 01/11/2008

Dernière mise à jour : 19/03/2010

Cette FAQ ne prétend pas traiter de tous les trucs & astuces ou de tous les problèmes rencontrés par les utilisateurs de WinDev : elle a été créée via l'analyse des questions les plus fréquentes posées sur le forum de ce site, et sur les connaissances des différents rédacteurs. Nous espérons voir le nombre de questions/réponses augmenter régulièrement, pour couvrir le plus de thèmes possibles. Pour y participer, suivez le mode d'emploi donné dans la 2ème question !

## **Ont contribué à cette FAQ :**

Tunidev - Emmanuel Lecoester - Stéphane Bertolotti -  
Michel HERRSCHER - Eric ROUMEGOU - Romain  
PETIT - =JBO= - Dominique Daussy - Tanguy Pruvot -

1. Présentation de la FAQ WinDev (4) .....	4
2. Concepts, WLangage et environnement (65) .....	5
2.1. Concepts (8) .....	6
2.2. Aide WinDev (4) .....	8
2.3. W-Langage (11) .....	10
2.4. Environnement de développement (2) .....	14
2.5. Interface (12) .....	15
2.5.1. Fenetres (6) .....	16
2.5.2. Tables (3) .....	18
2.5.3. Divers (3) .....	19
2.6. Graphisme (0) .....	20
2.7. Système (6) .....	21
2.7.1. Fichiers (1) .....	22
2.7.2. Réseaux (1) .....	23
2.8. Bases de données (4) .....	26
2.9. Documentation et installation (1) .....	28
2.10. Divers (17) .....	29
2.10.1. Routines (7) .....	30
2.10.2. Manipulation de dates (6) .....	33
2.10.3. Divers (4) .....	36

## Sommaire > Présentation de la FAQ WinDev

### Guide de bon usage de la FAQ et du Forum

**Auteurs :** Emmanuel Lecoester ,

Lorsque vous vous trouvez devant un problème, voici la marche à suivre :

1/ Chercher dans l'aide de WinDev (F1) ou avec la soucoupe (Aide : Index soucoupe)

2/ Chercher dans la FAQ

3/ Chercher sur un moteur de recherche (par exemple Google)

4/ Faire une recherche sur le forum concerné.

5/ Si vous n'avez toujours pas trouvé de réponse à votre question : vous pouvez, après avoir lu les Règles du forum poster une question précise avec un titre clair et vous enrichirez ainsi la base de connaissances. Les messages concernant spécifiquement WinDev doivent être postés dans le **forum WinDev**

### Comment participer à l'amélioration de la FAQ WinDev ?

**Auteurs :** Emmanuel Lecoester ,

Le plus simple est de prendre contact avec le coordinateur des rédacteurs. Actuellement, il s'agit de **elecoest** pour les sujets concernant cette FAQ.

### Que faire si je remarque une erreur dans la FAQ?

**Auteurs :** Emmanuel Lecoester ,

Si vous constatez une coquille dans la FAQ, nous vous remercions par avance de la signaler au responsable de la faq: actuellement **elecoest**.

Attention, il ne répondra à aucune question technique si l'une des manipulation technique ne marchait pas sur votre configuration, sauf si l'erreur fait suite à une coquille dans la Question-Réponse.

### Remerciements !

**Auteurs :** Emmanuel Lecoester ,

Nous tenons à remercier tous les rédacteurs et correcteurs de cette Faq sans cesse en évolution.

Sommaire > Concepts, WLangage et environnement

Sommaire > Concepts, WLangage et environnement > Concepts

## WinDev, WebDev, WinDev Mobile... C'est quoi ?

Auteurs : [Emmanuel Lecoester](#) ,

WINDEV 14 inclut en standard, entre autre, les modules suivants : Editeur d'analyses UML et Merise, RAD, Générateur de Dossier, IDE, générateur d'IHM, L5G, Editeur de code, Générateur d'états PDF et Codes Barres, Gestion des versions, Débogueur, Groupware, Lien avec les langages tierces, Base de Données intégrée, lien avec toutes les Bases de Données, Liaison série RS-232, XML, Lien Word et Excel, Grapheur, Support des services WEB .NET et J2EE, Composants, Gestion des images, sons et vidéos, Ouverture aux sockets, API, DLL..., Multilangue, Générateur d'aide, Générateur d'Installation, Editeur de tests de non régression, Outil de maintenance automatique, ... Toutes les informations concernant WinDev sont sur le site de [PC-Soft](#).

## Où trouver de l'aide ?

Auteurs : [Emmanuel Lecoester](#) ,

Toutes les informations concernant WinDev sont sur le site de [PC-Soft](#). Vous y trouverez la description des produits, des cours, des vidéos et même une version "test" librement téléchargeable. De plus, la version commerciale est fournie avec un cours d'autoformation, des exemples et une aide complète.

## Migration

Auteurs : [Emmanuel Lecoester](#) ,

Pour les projets des versions antérieures à 7.5, il faut utiliser l'[outil de migration](#). Ensuite, les sources sont récupérables par toutes les versions existant à ce jour (WD12)

## Ressources

Auteurs : [Emmanuel Lecoester](#) ,

Liste des documents sur WinDev, WebDev et WinDev Mobile  
Vidéos techniques : [pour comprendre vite](#)  
Une série de cours Windev mis à votre disposition par [Jean-Luc Baptiste](#).

## Clé USB non reconnue

Auteurs : [Emmanuel Lecoester](#) ,

En installant une nouvelle machine, je constate que ma clé USB n'est pas reconnue. Pourtant tout a été bien contrôlé :  
\* La petite lumière est allumée, \* La clé marche sur un autre poste, \* Hinstall.exe version 4.95 est installé. La solution  
La clé a été branchée trop tôt (avant de lancer hinstall.exe), du coup, XP n'a pas reconnu le périphérique. Pour rétablir

la situation, il faut appliquer la méthode suivante et dans cet ordre précis. \* Enlever la clé USB \* Désinstaller le pilote avec : `Hinstall /R` \* Réinstaller le pilote avec : `Hinstall /i` \* Rebrancher la clé USB

### Comment Associer les extensions des fichiers de vos projets à l'Explorateur de Windows ?

Auteurs : Emmanuel Lecoester ,

Issu du blog du ST

Lors de l'installation de WINDEV, WEBDEV ou WINDEV Mobile, les extensions des fichiers de vos projets sont automatiquement associées à leurs applications respectives. Ainsi par exemple un double-clic dans l'Explorateur sur un fichier de projet ".WDP", fera le lancement de WINDEV. Un double-clic sur un fichier de données .FIC, fera le lancement de la visionneuse WDMAP... Si pour une raison quelconque ces associations de l'Explorateur sont supprimées (restauration système, installation d'autres programmes...), il est possible très simplement de les recréer grâce à l'utilitaire WShellReset.exe inclus avec WINDEV, WEBDEV et WINDEV Mobile

Son utilisation est des plus simple, lancement par l'Explorateur à partir du dossier \Programmes\, et validation du message de confirmation. Rien d'autre à faire, les extensions seront toutes en place après un redémarrage du poste !

### Existe il une façon de faire en sorte que les logiciels anti-spyware et autres ne détectent pas les applications réalisées à l'aide de Windev comme étant une menace élevée ?

Auteurs : Michel HERRSCHER ,

Oui, les déclarer comme logiciels approuvés! Mais on devra le faire à chaque nouvelle version du logiciel Windev. Autre chose que je fais, je déclare les extensions des data ( FIC NDX MMO) comme ne devant pas être scanné par les antivirus. Je sais que c'est un risque mais j'ai vu dans un message qu'un antivirus avait pris des fichiers data en otage ... donc, je préfère prévenir.

### Quelles sont les différences entre la version express et la version complète ?

Auteurs : Emmanuel Lecoester ,

l'éditeur présente la liste des restriction sur cette page : [Source](#) Limitations.

Sommaire > Concepts, WLangage et environnement > Aide WinDev


### Comment accéder à l'aide en ligne ?

Auteurs : =JBO= ,

Elle se trouve au bout de ce lien:  <http://doc.pcsoft.fr/fr-FR/>

### Comment accéder à l'aide locale ?

Auteurs : =JBO= ,

Voici la manipulation : Outils / Options / Options générales de Windev / Onglet Aide en cochant l'option "utiliser la base d'aide locale." Le tout en image  [ici](#)

### Affiner le résultat des recherches dans l'aide en ligne

Auteurs : =JBO= ,

Dans l'aide en ligne internet, la zone de recherche est à double usage:

- recherche par index,
- recherche par mot-clé et expression;

Vous êtes habitué à la recherche par index: au fur et à mesure de votre saisie, une liste (ici un popup) vous présente des réponses possibles, sur lesquels on peut cliquer pour afficher la page correspondante. Mais parfois, on affiche une page qui présente un ensemble de résultats plus ou moins pertinents, et parfois une seule page ne suffit pas à les afficher tous ! Pour mieux cibler l'information recherchée, et affiner l'ensemble des résultats, l'aide en ligne fournie par PC Soft offre un outil de recherche intégrée. Il permet d'effectuer une recherche sur plusieurs mots clés et aussi sur des expressions. Une expression est un ensemble de mots placés entre guillemets et qui seront recherchés à l'identique.

Voici quelques exemples:

- La recherche sur les mots clés [requête intégrée] ramène 21 résultats.
- La recherche sur l'expression ["requête intégrée"] ramène 13 résultats.
- La recherche sur les mots clés [HExécuteRequête requête intégrée] ramène 4 résultats.
- La recherche sur le mot clé + l'expression [HExécuteRequête "requête intégrée"] ramène 2 résultats.

### Annoter l'aide en ligne

Auteurs : =JBO= ,

Vous savez que nous pouvons annoter les pages de la documentation en ligne. Cependant, la recherche intégrée de la documentation en ligne ne tient pas compte des annotations. Si vous souhaitez faire une recherche sur l'intégralité des pages de la documentation en ligne, c'est quand même possible grâce au moteur de recherche Google. Celui-ci propose de limiter une recherche à un site web précis, grâce à une instruction spécifique que vous ajoutez après vos mots-clés et expressions recherchées:

```
site:"adresse site web"
```

Pour cibler la documentation en ligne de PC Soft ce sera:

```
site:http://doc.pcsoft.fr/
```

En pratique, si on effectue une recherche sur le mot clé [polymorphisme]...

- La recherche intégrée ramène 3 résultats.
- La recherche Google propose 5 résultats (mais elle fait l'impasse sur 1 des résultats ramenés par la recherche intégrée )

Pour effectuer cette recherche sur Google j'ai donc tapé ceci:

```
polymorphisme site:http://doc.pcsoft.fr/
```

Il faut envisager la recherche avec Google comme un complément à la recherche intégrée; elle ne la remplacera pas. En effet, la recherche avec Google est excellente mais elle n'est pas parfaite:

- elle ramène des résultats doublons,
- elle ignore certaines pages vers lesquelles ne mène aucun lien hypertexte,
- elle n'est pas instantanément mise à jour quand la documentation en ligne est modifiée.

Sommaire > Concepts, WLangage et environnement > W-Langage

## Comment copier un Objet ?

Auteurs : **Emmanuel Lecoester** ,

**Pour faire une copie d'objet il suffit de faire une simple affectation :**

```
Objet1 est un Classe
Objet2 est un Classe

Objet2 = Objet1
```

## Appel du constructeur de la classe ancêtre depuis une classe dérivée

Auteurs : **Emmanuel Lecoester** ,

**Exemple avec le constructeur d'une classe dérivée :**

```
PROCEDURE Constructeur(Param1,Param2)
//appel du constructeur de la classe ancêtre (si nécessaire)
//Si le constructeur de la classe ancêtre n'a pas de paramètre
//L'appel est automatique, il n'y a rien à faire
//A mettre impérativement au début avant les autres initialisations
Ancêtre:Constructeur(Param1)
//OU
//Constructeur Ancêtre(Param1)
//OU
//En cas d'héritage multiple, ou pou être plus explicite on peut nommer la classe ancêtre :
//ClasseAncêtre:Constructeur(Param1)
//OU
//Constructeur ClasseAncêtre(Param1)
```

## Comment savoir si un objet dynamique est alloué ?

Auteurs : **Emmanuel Lecoester** ,

**Il suffit de le comparer avec NULL.**

**Exemple :**

```
//Objet1 est un objet dynamique
Objet1 est un objet dynamique

//Traitement qui alloue ou non l'objet

//Ici comment savoir si l'objet Objet1 est alloué ?
SI Objet1=NULL ALORS
//On est dans ce cas : Objet non alloué
//Ici on alloue l'objet
Objet1 = allouer un objet MaClasse
SINON
//on est dans le cas : Objet déjà alloué
```

```
FIN
```

## Gérer l'indirection sur un onglet

**Auteurs :** Emmanuel Lecoester ,

```
NomFenetreAffichée est une chaîne = "MaFenetre"  
champ_onglets_en_cours est une chaîne = "MonOnglet"  
volet_en_cours est un entier = 1
```

### La commande

```
{NomFenetreAffichée+"."+champ_onglets_en_cours, indChamp}[volet_en_cours]..Visible = Vrai
```

est équivalente à

```
MaFenetre.MonOnglet[4]..Visible = Vrai
```

## Comment utiliser une indirection comme propriété

**Auteurs :** Emmanuel Lecoester ,

Les indirections sont très intéressantes avec quelques limitations dont celle-ci : les indirections ne sont pas utilisables pour des accès à une propriété

```
prop est une chaîne = "Libellé"  
BTSEL..{prop}="Rechercher"
```

Pour contourner cela rien ne vaut de passer par la compilation dynamique :

```
Compile("traduction", "BTSEL.."+prop+"="+"Rechercher""")  
ExécuteTraitement("traduction", trtProcédure)
```

## Comment ajouter un guillemet dans une chaîne ?

**Auteurs :** Emmanuel Lecoester ,

soit en doublant les doubles quotes, soit en ajoutant le caractère ascii correspondant

```
Info("Bonjour ""le monde""")
```

```
Info("Bonjour "+caract(34)+"le monde"+caract(34))
```

## Comment récupérer les arguments de la ligne de commande passée à l'exécutable ?

Auteurs : **Emmanuel Lecoester** ,

```
Résultat = LigneCommande() renvoie toute la ligne de commande  
Résultat = LigneCommande(indice) renvoie le contenu du paramètre ayant pour indice  
Résultat = LigneCommande(Nom du paramètre, Valeur par défaut) si le paramètre demandé n'existe pas,  
c'est la valeur par défaut qui est retournée
```

## Optimisation des boucles POUR

Auteurs : **Emmanuel Lecoester** ,

Qui n'a jamais fait une boucle classique utilisant le nombre de lignes d'une table mémoire, par exemple :

```
POUR i=1 A TableOccurrence(TABLE1)  
// traitement  
FIN
```

Depuis plusieurs versions, vous aviez donc comme moi, une information indiquant que cette syntaxe n'est pas conseillée car il est préférable de ne pas réévaluer à chaque fois le nombre de lignes et donc d'utiliser une variable avant, ce qui me semble judicieux. Donc, j'ai fait tout comme on me le suggérait :

```
nLignes est un entier = TableOccurrence(TABLE1)  
POUR i=1 A nLignes  
// traitement  
FIN
```

et bien, maintenant il est possible de faire beaucoup plus simple en utilisant le mot clé "\_A\_" dans la boucle (un peu comme les mots clés \_ET\_ et \_OU\_) qui permet de faire la même optimisation, donc en économisant la variable temporaire.

```
POUR i=1 _A_ TableOccurrence(TABLE1)  
// traitement  
FIN
```

## Astuce sur l'instruction SELON

Auteurs : **Emmanuel Lecoester** ,

L'instruction conditionnelle "SELON" permet de choisir d'exécuter une action ou une autre en fonction de la valeur d'une expression. Il est possible de tester une valeur comprise entre un minimum et un maximum, grâce à l'opérateur : **min <\*> max.**

```
SELON nHeure  
CAS 0<*<16>("Bonjour")  
CAS 16<*<23>("Bonsoir")
```

FIN

## Impression par page

**Auteurs : Emmanuel Lecoester ,**

**WinDev, WebDev et WinDev Mobile proposent la propriété**

```
..ImpressionAChaquePage
```

Cette propriété, très utile, découpe l'impression d'un état page par page. Cela permet par exemple de minimiser la taille du spooler d'impression ou d'envoyer les pages de l'état sur des bacs différents.

Vous trouverez tous les détails de cette propriété dans l'aide en ligne.

## Faire une fonction EstNumérique

**Auteurs : Emmanuel Lecoester ,**

**Avec un point comme séparateur**

```
VérifieExpressionRégulière(Saisie1, "[0-9]+[.][0-9]+")
```

**Avec une virgule comme séparateur**

```
VérifieExpressionRégulière(Saisie1, "[0-9]+|([0-9]+[,][0-9]+)")
```

Sommaire > Concepts, WLangage et environnement > Environnement de développement

### WinDev 14 dans la base de registre

**Auteurs : Emmanuel Lecoester ,**

**Avec Regedit il suffit de supprimer ou de renommer la clé.**

```
HKEY_CURRENT_USER\Software\PC SOFT\WinDev\14.0
```

**Au lancement suivant, WinDev la recrée automatiquement en restaurant toutes les valeurs par défaut. Il considère que c'est sa première utilisation.**

### Comment redéfinir l'aide locale comme aide par défaut dans WinDev 15?

**Auteurs : Emmanuel Lecoester ,**

**Outils / Options / Options générales de Windev / Onglet Aide en cochant l'option "utiliser la base d'aide locale.**

Sommaire > Concepts, WLangage et environnement > Interface

Sommaire > Concepts, WLangage et environnement > Interface > Fenetres

Comment configurer l'effet de transition entre les onglets d'un champ onglet ?

**Auteurs :** Emmanuel Lecoester ,

Tout est dans : [Projet](#) / [Description du projet](#) / [Avancé](#) / [Animation des champs](#)

Connaitre la position de la souris sur un contrôle précis

**Auteurs :** Dominique Daussy ,

```
PosSouris est compose de
  HHH est un entier long
  VVV est un entier long
FIN
AppelDLL32("user32","GetCursorPos" , &PosSouris)
AppelDLL32("user32","ScreenToClient" , handle("NOMDUCONTROLE"), &PosSouris)
Trace("Zone en "+PosSouris.HHH+", "+PosSouris.VVV)
```

Détecter le mot sous la souris dans un champ de saisie

**Auteurs :** Dominique Daussy ,

```
LParam est un entier = _EVE.lParam
LOC_Res est un entier
LOC_Mot est une chaîne
LOC_Pos1,LOC_Pos2 sont des entiers
LOC_TexteRTF est une chaîne

LOC_Res = SendMessage(Handle(Saisie1..Nom),215,0,LParam)
LOC_Pos1 = Position(Saisie1," ",LoWord(LOC_Res),DepuisFin)+1
LOC_Pos2 = Position(Saisie1," ",LOC_Pos1)-1

Trace(Saisie1[[LOC_Pos1 A LOC_Pos2]])
```

Comment récupérer le nom des champs d'une fenêtre ?

**Auteurs :** Emmanuel Lecoester ,

```
// Remplir une combo avec la liste des champs de la fenêtre
i est un entier = 1
ResChamp est une chaîne
ResChamp = EnumereChamp(FenetreSaisie, i)
TANTQUE ResChamp<>" "
  i++
  ListeAjoute(ListeChamp, ResChamp)
ResChamp = EnumereChamp(FenetreSaisie, i)
```

FIN

### Comment mettre le titre à droite et à gauche

**Auteurs :** Emmanuel Lecoester ,

```
TitreEnCours(EOT+gCadrage("G","partie gauche")+gCadrage("D","partie droite"))
```

### Comment mettre une fenetre au dessus des autres fenetres?

**Auteurs :** Emmanuel Lecoester ,

```
FenêtreDessusTout(Handle(MaFenêtre..Nom))
```

avec comme code :

```
API("user32","SetWindowPos",nHandleFenêtre,-1,0,0,0,0,3)
```

Sommaire > Concepts, WLangage et environnement > Interface > Tables

colorier une ligne dans une table en fonction de la valeur d'une colonne (macolonne)

Auteurs : Emmanuel Lecoester ,

Il faut mettre le code suivant dans la section "Affichage d'une ligne de la table".

```
si Table.macolonne[3] ALORS Table[3]..CouleurFond = CouleurRouge
```

Récupérer le libellé de toutes les rubriques d'un fichier à l'intérieur d'une boucle TANTQUE pour l'insérer dans une table hiérarchique

Auteurs : Emmanuel Lecoester ,

La table contient 2 Colonnes - 1 : Col\_NomFichier - 2 : COL\_Parent

```
LOCAL
sNomFich, sListeRubq, sRubq, sLibRubq sont des chaînes
nParent est un entier

sNomFich = "LigneMvt"

// [ Récupérer les rubriques du fichier ]
sListeRubq = HListeRubrique( sNomFich,hLstDétailPlus)

// [ Racine ]
nParent = TableAjouteLigne(TABLEH_Fichier, sNomFich)

// [ Récupérer les libellé des rubriques ]
POUR TOUTE CHAINE sRubq DE sListeRubq SEPAREE PAR RC

    // [ Libellé de la rubrique ]
    sLibRubq = ExtraitChaîne(sRubq,6)

// [ Ajoute le libellé ]
TABLEH_Fichier..ImageEnroulée = IMG_Fichier // ->
TableAjouteFils(TABLEH_Fichier, nParent, sLibRubq)
FIN
```

Gestion des totaux dans une table mémoire.

Auteurs : Eric ROUMEGOU ,

J'ai une table mémoire que je remplis avec une requête. J'utilise les fn de champs calculés pour avoir les sommes sur certaines colonnes de cette table.

Les sommes fonctionnent bien mais puis je récupérer directement les valeurs de ces champs calculés ? Voici la solution:

```
MaColonne[ligneTotal]..valeur
MaColonne[ligneMoyenne]..couleurfond
MaColonne[ligneComptage]..etc
```

Sommaire > Concepts, WLangage et environnement > Divers

### Comment se positionner sur un onglet déterminé dans une feuille excel ?

**Auteurs : Emmanuel Lecoester ,**

```
gMondocument est un xlsDocument
gMondocument = xlsOuvre("test.xls")

FONCTION PositionnerOnglet(sNomOnglet est une chaîne)
n est un entier

POUR n = 1 A xlsNbFeuille(gMondocument)
xlsFeuilleEnCours(gMondocument,n)
SI sNomOnglet = gMondocument.NomFeuille ALORS RENVOYER Vrai
FIN
RENOYER Faux
```

### Comment détecter l'appui sur la touche ALT ?

**Auteurs : Emmanuel Lecoester ,**

**La touche ALT n'est pas une touche en tant que tel, c'est une touche système. Il faut placer le code :**

```
ToucheEnfoncée(teAlt)
```

**dans un traitement optionnel qui s'appelle WM\_SYSKEYDOWN.**

### Comment trouver les identifiants des éléments sélectionnés dans un champ liste multisélection ?

**Auteurs : Romain PETIT ,**

**La fonction ListeSelect est là pour ça :**

```
i est un entier = 1
LigneSelectionnee est un entier
LigneSelectionnee = ListeSelect(Liste_Stock,1)
TANTQUE LigneSelectionnee <= Liste_Stock.Count
    Trace("La ligne n°"+ LigneSelectionnee +" est sélectionnée ->" + Liste_Stock[LigneSelectionnee])
    i++
    LigneSelectionnee = ListeSelect(Liste_Stock,i)
FIN
```

Sommaire > Concepts, WLangage et environnement > Graphisme

Sommaire > Concepts, WLangage et environnement > Système

Sommaire > Concepts, WLangage et environnement > Système > Fichiers

## Enregistrer un fichier texte en UTF8 ?

**Auteurs : Emmanuel Lecoester ,**

`ChaîneVersUTF8`

Sommaire > Concepts, WLangage et environnement > Système > Réseaux

Obtenir l'adresse MAC à partir d'une adresse IP

Auteurs : **Romain PETIT**,

```

FUNCTION getAdresseMACDistante(p_szAdresseIP=NetAdresseIP(),p_szErr="")

// Version 1.00 pour WD7 (WD55 voir les déclarations)
// (c) Octobre 2002 Romain PETIT (mailto:rompetit@ifrance.com)
// sous licence WD-LIBRE (http://www.windevasso.org)
// But : obtenir l'adresse MAC à partir d'une adresse IP (réseau local)
// ATTENTION, ne fonctionne que sous 2K/XP (nécessite le SP6 sous NT ?)
// Paramètres : adresse IP au format xx.xx.xx.xx
// Retour : chaine, adresse MAC (6 nombres en hexa séparés par un '-')
// en cas d'erreur, la fonction retourne une chaine vide et la chaine sErr passée par adresse peut
être récupérée.
// Notes : il ne s'agit pas ici de récupérer l'adresse MAC de machines sur Internet
// mais de machines sur un réseau local (non testé avec routeurs).
// Dans le cas d'une connexion ADSL Ethernet, NetAdresseIP() renvoie l'adresse
// attribuée par le provider, la fonction renvoie donc l'adresse MAC de la
// machine établissant la connexion (modem Ethernet par exemple)
// Références API:
// http://msdn.microsoft.com/library/default.asp?url=/library/en-us/iphlp/iphlp/sendarp.asp
// Exemple :
// sAdrMAC est une chaine
// sRetourErreur est une chaine=""
// sIPDist est une chaine = "192.168.0.1"
// sAdrMAC = fg_AdresseMACDistante(sIPDist, sRetourErreur)
// SI sAdrMAC <>" ALORS
// info("L'adresse MAC de "+ sIPDist + " est "+sAdrMAC)
// SINON
// erreur("L'adresse MAC de "+ sIPDist + " n'a pas pu être obtenue : "+ sRetourErreur)
// FIN

SI p_szAdresseIP ~="" ALORS p_szAdresseIP = NetAdresseIP()
SI SysVersionWindows(sysVersionPlateForme)<>"NT" ALORS
  p_szErr = "Cette fonction n'est pas disponible avec la version de Windows installée."
  RENVOYER ""
FIN
szAdresseIP est une chaîne ASCIIZ sur 255 = p_szAdresseIP
lAdresseIP est un entier sans signe
lAdresseMAC est un entier sans signe
tlAdresseMAC est un tableau fixe de 6 entiers sans signe sur 1 octet
lNbAdrPhys est un entier sans signe
lRet est un entier
i est un entier
szRetour est une chaîne=""

// Conversion de l'adresse IP en entier long
lAdresseIP = AppelDLL32("wsock32.dll", "inet_addr", &szAdresseIP)
SI lAdresseIP <> 0 ALORS
  lNbAdrPhys = 6
  //récupération de l'adresse MAC
  lRet = AppelDLL32("iphlpapi.dll", "SendARP", lAdresseIP, 0,&lAdresseMAC, &lNbAdrPhys)
  SI lRet=0 ALORS
    SI lAdresseMAC<> 0 ET lNbAdrPhys<> 0 ALORS
      //La valeur retournée est un pointeur sur un tableau
      //que l'on copie dans le tableau de 6 octets prévu
      Transfert(&tlAdresseMAC, &lAdresseMAC, 6)//6*1 octet
      //On peut utiliser : AppelDLL32("kernel32","RtlMoveMemory",&tlAdresseMAC,&lAdresseMAC,6)
      //Pour chaque élément du tableau, on reconstruit la chaine en hexa
      POUR i=1 A lNbAdrPhys
        szRetour += NumériqueVersChaîne(tlAdresseMAC[i],"02X")+ "-"
      FIN
      szRetour = Gauche(szRetour, Taille(szRetour)-1)
      p_szErr = ""
    
```

```
SINON
  p_szErr = "La récupération de l'adresse MAC à partir de l'IP "+ p_szAdresseIP+" a échoué."
FIN
SINON
  p_szErr = "La récupération de l'adresse MAC à partir de l'IP "+ p_szAdresseIP+" a échoué."
FIN
SINON
  p_szErr = "L'adresse IP "+p_szAdresseIP+" n'a pas pu être convertie. Vérifiez le format."
FIN
RENOYER szRetour
```

## Comment connaître le numéro du disque dur ?

**Auteurs : Emmanuel Lecoester ,**

```
fdisqueinfo("c:", fdnumerodeserie)
```

## Comment reconnaître si le système est français ou anglo-saxon ?

**Auteurs : Emmanuel Lecoester ,**

```
// To detect if the host computer uses the french language
// Détermine si le français est la langue utilisée sur l'ordinateur
FUNCTION IsFrench()
bRet is int = False
sLocale is fixed string on 6
API("KERNEL32", "GetLocaleInfoA", LOCALE_USER_DEFAULT, LOCALE_SENGLANGUAGE,
& amp; amp; amp; amp; sLocale, Length(sLocale))
IF Upper(sLocale) = "FRENCH" THEN bRet = True
RESULT bRet
```

## Comment récupérer la langue par défaut de Windows ?

**Auteurs : Emmanuel Lecoester ,**

```
i est un entier système
clangue est une chaîne fixe sur 100
i=API("kernel32","GetSystemDefaultLangID")
i=API("kernel32","VerLanguageNameA",i, &clangue,100)
SI i <>0 ALORS Trace(clangue,i)
```

## Comment changer par programmation l'icône d'une fenêtre ?

**Auteurs : Emmanuel Lecoester ,**

```
//-----
//Il suffit d'initialiser sPath au chemin désiré...
sPath est chaîne = "C:\"

SHFILEINFOA est une structure
hIcon est un entier système //Type C : HICON
iIcon est un entier //Type C : int
dwAttributes est un entier //Type C : DWORD
szDisplayName est une chaîne fixe sur 260 //Type C : CHAR
```

```
szTypeName est une chaîne fixe sur 80 //Type C : CHAR
FIN
stFileInfo est un SHFILEINFO

//SHGFI_SMALLICON = 1
//SHGFI_SYSICONINDEX = 16384
nSHGFI est entier = SHGFI_SMALLICON | SHGFI_SYSICONINDEX
//Handle de l'imageList
nIMAGELIST_16 est entier = API("Shell32.dll", "SHGetFileInfo", "", 0, &stFileInfo,
    Dimension(stFileInfo), nSHGFI)
//Index de l'icone dans l'imageList
API("Shell32.dll", "SHGetFileInfo", sPath, 0, &stFileInfo, Dimension(stFileInfo), nSHGFI)
nIconIndex est entier = stFileInfo:iIcon
//Handle de l'icone
//ILD_NORMAL = 0
//ILD_TRANSPARENT = 1
nHIcon est entier = API("COMCTL32.DLL", "ImageList_GetIcon", nIMAGELIST_16, nIconIndex, ILD_NORMAL |
    ILD_TRANSPARENT)
//Mise à jour de l'icone de la fenêtre
//WM_SETICON = 128
//ICON_SMALL = 0
SendMessage(Handle(MaFenêtre), WM_SETICON, ICON_SMALL, nHIcon)
//Ne pas détruire l'icone, sinon l'affichage ne sera pas réalisé
//API("user32", "DestroyIcon", nHIcon)
//-----
```

Sommaire > Concepts, WLangage et environnement > Bases de données

### Créer les fichiers HyperFile seulement si ils n'existent pas

Auteurs : Emmanuel Lecoester ,

```
HCréationSiInexistant("")
HGèreIntégrité("", "", hCardinalité+hEnModification+hEnSuppression, Faux)
```

### Filtrer sur deux rubriques

Auteurs : Emmanuel Lecoester ,

```
hfiltrer(fichier, "rubrique1=' "+champ+" ' et rubrique2=' "+champ2+" ' ")
```

### Simuler l'ordre "insert into" d'une table à une autre

Auteurs : Emmanuel Lecoester ,

```
PROCEDURE INSERTINTO(sTableSource est une chaîne, sTableCible est une chaîne, MDPSOURCE est une chaîne
="", sMDPCIBLE est une chaîne="")

ListeRubriqueSource, ListeRubriqueCible est une chaîne
ReqClient est une Source de Données
tot est un entier
HDéclareExterne(sTableSource, "SOURCEWD", MDPSOURCE)
HDéclareExterne(sTableCible, "CIBLEWD", sMDPCIBLE)

HOuvre("SOURCEWD", MDPSOURCE, hOlectureEcriture)
HOuvre("CIBLEWD", sMDPCIBLE, hOlectureEcriture)

HLitPremier("SOURCEWD", "", hBlocageEcriture)

ListeRubriqueSource=HListeRubrique("SOURCEWD")
ListeRubriqueCible=HListeRubrique("CIBLEWD")

TANTQUE HTrouve("SOURCEWD")
  POUR i=1 _A_ ChaîneOccurrence(ListeRubriqueSource, RC)+1
    HVersRubrique("CIBLEWD", i, HRécupèreRubrique("SOURCEWD", i))
  FIN
  HAjoute("CIBLEWD")
  HLitSuivant("SOURCEWD")
FIN
HAnnuleDéclaration("SOURCEWD")
HAnnuleDéclaration("CIBLEWD")
StephR1 sur tunidev
```

### Comment lire le contenu d'une base de données ACCESS avec WinDev ?

Auteurs : Emmanuel Lecoester ,

Pour récupérer la liste des tables composant une base Access, il suffit d'utiliser le code ci-dessous :

```
// Variables
ListeFichier est une chaîne
```

```
ListeFichier=HListeFichier("", "", "c:\base.mdb", hOleDbAccess2007, hOLectureEcriture)
```

Sommaire > Concepts, WLangage et environnement > Documentation et installation

**Y'a t'il un moyen d'inclure les fichiers de données lors de la compilation du projet ?**

**Auteurs : Emmanuel Lecoester ,**

**A la page 4 "Mise en bibliothèque" de l'assistant de création de l'exécutable, il y a un bouton "Ajouter" qui ouvre le sélecteur de fichier où on peut sélectionner et importer dans la bibliothèque de l'application les fichiers souhaités.**

Sommaire > Concepts, WLangage et environnement > Divers

Sommaire > Concepts, WLangage et environnement > Divers > Routines

### Calculer un hash MD5

Auteurs : **Eric ROUMEGOU** ,

```
PROCEDURE MonCalculMd5(pCh)
wch,Pwdmd5 est chaîne
wch=HashChaîne(HA_MD5_128,pCh)
i,imax est un entier
Pwdmd5=""
imax=Taille(wch)
POUR i=1 A imax
    Pwdmd5+=NumériqueVersChaîne(Asc(wch[[i]]), "02X")
FIN
Pwdmd5=Minuscule(Pwdmd5)
RENOYER Pwdmd5
```

### Récupération des processus en cours

Auteurs : **Emmanuel Lecoester** ,

Tout simplement avec la fonction WinDev ExeListeProcessus

### WDMoFic par programmation

Auteurs : **Emmanuel Lecoester** ,

Voici un petit code à placer dans l'initialisation de vos projets pour mettre à jour automatiquement vos fichiers de données. Il faut bien sûr préalablement intégrer l'outil "WDMoFic.exe" avec toutes ses dépendances (DLL) ainsi que le ".wdd" de votre application dans le répertoire d'installation du projet.

```
bResultat est un booléen=Faux
QUAND EXCEPTION DANS
    SI PAS HCréationSiInexistant("")ALORS Erreur("Impossible de créer les fichiers !")
FAIRE
    bResultat=LanceAppli(""+fRepEnCours()+"\WDMoFic.exe" -error -nobackup "+"/
WDD=""+fRepEnCours()+"\Monanalyse.wdd ""+"/DIR=""+fRepExe+"",exeBloquant)
    Info("Le logiciel à détecté une incompatibilité de vos fichiers de données. Votre base de données
a été modifiée et mise à jour automatiquement."+RC+"Veuillez relancer le logiciel SVP... ")
    FinProgramme()
FIN
```

### Comment connaître le navigateur par défaut

Auteurs : **Emmanuel Lecoester** ,

```
navigateur est une chaîne
nomproduit est une chaîne
navigateur =RegistreLit("HKEY_CLASSES_ROOT\HTTP\Shell\open\command",1)
navigateur=fExtraitChemin(navigateur,fDisque+fRépertoire+fFichier)+"exe"
nomproduit=ExeInfo("ProductName",navigateur)
SI nomproduit="" ALORS
    nomproduit="Internet Explorer"
FIN
```

```
Info(nomproduit)
```

## Afficher la boîte de dialogue "Ouvrir Avec"

**Auteurs : Emmanuel Lecoester ,**

```
LanceAppli("rundll32.exe shell32.dll, OpenAs_RunDLL " + NomFichier)
```

## Dessiner l'icône associée à une extension de fichier

**Auteurs : Tanguy Pruvot ,**

```
// Résumé : Dessine l'icône associée à une extension de fichier, ou d'un fichier en particulier
// Paramètres :
// sFileName (chaîne) : Nom du fichier ou extension avec le "." (ex: ".doc")
// sNomChpImage (chaîne - valeur par défaut="") : Nom du Champ Image
// hDC (entier système - valeur par défaut=0) : ou hDC du champ image renvoyé par dDebutDessin()
// nTaille (entier - valeur par défaut=16) : 16, 32, 48, 64
// Valeur de retour :
// booléen
//
// Exemples :
// DessineIcôneFichier(".doc","IMG_Icône")
// DessineIcôneFichier("C:\Test.exe","",hDC)

PROCEDURE DessineIcôneFichier(sFileName est chaîne, sNomChpImage est chaîne="", ...
    LOCAL hDC est entier système=0, nTaille est entier = 16)

SI sNomChpImage<>" _ET_ PAS ChampExiste(sNomChpImage) ALORS
    RENVOYER Faux
FIN

bRes est booléen

sDefaultIcon est chaîne = SysRep(srSystème) + "\SHELL32.DLL"
nIconIndex est entier
sFileExt est chaîne
sProgramName est chaîne
nbIcons,nIndex,nError sont entier
hIcon est entier système

sFileExt = Minuscule(fExtraitChemin(sFileName,fExtension))

SELON sFileExt
CAS ".ico"
    SI fFichierExiste(sFileName) ALORS
        sDefaultIcon=sFileName
        nIconIndex = 0
        GOTO DRAW_ICON
    FIN
CAS ".exe"
    //sDefaultIcon = SysRep(srSystème) + "\SHELL32.DLL"
    nIconIndex = 2
    //Icône de l'exécutable (si nom de fichier complet fourni)
    SI fFichierExiste(sFileName) ALORS
        sDefaultIcon=sFileName
        nIconIndex = 0
    FIN
    GOTO DRAW_ICON
FIN
```

```
SI RegistreExiste("HKEY_CLASSES_ROOT\""+sFileExt) ALORS
sProgramName = RegistreLit("HKEY_CLASSES_ROOT\""+sFileExt,0)
sDefaultIcon = RegistreLit("HKEY_CLASSES_ROOT\""+sProgramName + "\DefaultIcon",0)

nIndex = Position(sDefaultIcon, ",", Taille(sDefaultIcon), DepuisFin)
SI nIndex ALORS
  sDefaultIcon = sDefaultIcon[[A nIndex-1]]
  nIconIndex = sDefaultIcon[[nIndex+1 A]]
FIN
FIN

DRAW_ICON:
SI nTaille=16 ALORS
  //Petite icone
  nbIcons = API("SHELL32","ExtractIconExA", sDefaultIcon, nIconIndex, Null, &hIcon, 1)
SINON
  //Icône standard
  hIcon = API("SHELL32","ExtractIconA", 0, sDefaultIcon, nIconIndex)
FIN

SI hIcon DANS (-1,0,1) ALORS
  RENVOYER Faux
FIN

SI sNomChpImage<>" " ALORS
  {sNomChpImage,indChamp}..Visible=Faux
  hDC = dDébutDessin(sNomChpImage)
  nError = API("USER32","DrawIconEx",hDC, 0, 0, hIcon, nTaille, nTaille, 0, 0, 3)
  {sNomChpImage,indChamp}..Visible=Vrai
SINON
  nError = API("USER32","DrawIconEx",hDC, 0, 0, hIcon, nTaille, nTaille, 0, 0, 3)
FIN

API("USER32","DestroyIcon",hIcon)

RENOYER Vrai
```

Je voudrais connaître le nombre de pages d'un fichier pdf mais sans devoir l'ouvrir !

Auteurs : Emmanuel Lecoester ,

La fonction WinDev PDFNombreDePages est là pour ça !

```
NbPages est un entier
NbPages = PDFNombreDePages("C:\MonDocument.pdf")
Info("Nombre de pages du PDF : " + NbPages)
```

[Sommaire](#) > [Concepts, WLangage et environnement](#) > [Divers](#) > [Manipulation de dates](#)

## Comment vérifier la validité d'une date

**Auteurs : Stéphane Bertolotti ,**

```
MaSemaine est un entier
MaSemaine = NumeroDeSemaine("20080916")
```

## Comment récupérer le jour d'une date

**Auteurs : Stéphane Bertolotti ,**

```
Monjour est un entier
Monjour = DateVersJour("20080916")
//Monjour = 2 dans ce cas
//Si le résultat de la fonction précédente est supérieur à 5 alors il s'agit d'un jour du week-end
```

## Comment ajouter un nombre de mois à une date ?

**Auteurs : Emmanuel Lecoester ,**

```
UneDate est une date
UneDate = DateSys()
UneDate..Mois += nombredemois
```

## Séparer le jour, le mois et l'année d'une date

**Auteurs : Emmanuel Lecoester ,**

```
MaDate est une Date = DateSys()
jour =Gauche(DateVersChaîne(MaDate),2)
mois =Milieu(DateVersChaîne(MaDate),4,2)
année=Milieu(DateVersChaîne(MaDate),7,4)
```

## Convertir une date RFC 822 en date Windev

**Auteurs : Dominique Daussy ,**

```
// Syntaxe :
//DateRFC822_VersDateHeure (Par_DateRFC822, ParVar_Date)
//
// Paramètres :
//     Par_DateRFC822 : Date chaine au format RFC822
//     ParVar_Date : Variable par adresse. Contendra le résultat de la conversion.
// Valeur de retour :
//     Aucune
//// Notes :
// Sur internet, vous rencontrerez souvent des dates au format RFC 822.
// Cette procédure permet de convertir une date de ce format vers le format Windev.
```

```

PROCEDURE DateRFC822_VersDateHeure(Par_DateRFC822,ParVar_Date)

LOC_Chaine est une chaîne = Par_DateRFC822
LOC_DateHeure est une DateHeure

// *****
// * ON SUPPRIME LE NOM DU JOUR *
// *****
SI Position(LOC_Chaine,",") ALORS
    LOC_Chaine=SansEspace(LOC_Chaine[[Position(LOC_Chaine,",")+1 A]])
FIN

// *****
// * ON REMPLACE LE NOM DU MOIS PAR SON NUMERO *
// *****
LOC_Chaine=Remplace(LOC_Chaine,"Jan","01",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Feb","02",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Mar","03",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Apr","04",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"May","05",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Jun","06",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Jul","07",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Aug","08",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Sep","09",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Oct","10",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Nov","11",SansCasse)
LOC_Chaine=Remplace(LOC_Chaine,"Dec","12",SansCasse)
LOC_DateHeure..PartieDate=ChaîneVersDate(LOC_Chaine,"JJ MM AAAA")
LOC_DateHeure..PartieHeure=Remplace(ExtraitChaîne(LOC_Chaine,4," "),":","")

ParVar_Date=LOC_DateHeure
    
```

## Comment convertir une date d'un calendrier gregorien en hégrorien ?

**Auteurs : Tunidev ,**

```

PROCEDURE ConversionMiladiHejri(DateMiladi est une Date)

SI DateMiladi ="" ALORS DateMiladi = DateSys()

y est un entier = DateMiladi..Année
m est un entier = DateMiladi..Mois
d est un entier = DateMiladi..Jour

jd est un entier

SI ((y > 1582) _OU_ (y = 1582 ET m > 10) _OU_ (y = 1582 ET m = 10 ET d > 14)) ALORS
jd=PartieEntière((1461*(y
+4800+PartieEntière((m-14)/12)))/4)+PartieEntière((367*(m-12*(PartieEntière((m-14)/12)))/12)-
PartieEntière((3*(PartieEntière((y+4900+PartieEntière((m-14)/12))/100)))/4)+d-32075
SINON
jd = 367*y-PartieEntière((7*(y+5001+PartieEntière((m-9)/7)))/4)+PartieEntière((275*m)/9)+d+1729777
FIN

l est un entier=jd-1948440+10632
n est un entier=PartieEntière((l-1)/10631)
l=l-10631*n+354
j est un entier=(PartieEntière((10985-
l)/5316))*(PartieEntière((50*l)/17719))+(PartieEntière(l/5670))*(PartieEntière((43*l)/15238))
l=l-(PartieEntière((30-j)/15))*(PartieEntière((17719*j)/50))-
(PartieEntière(j/16))*(PartieEntière((15238*j)/43))+29
m=PartieEntière((24*l)/709)
    
```

```
d=1-PartieEntière((709*m)/24)
y=30*n+j-30
```

HMonths\_L est une chaîne

SELON m

```
CAS 1 : HMonths_L = "Muharram"
CAS 2 : HMonths_L = "Safar"
CAS 3 : HMonths_L = "Rabi al-awwal"
CAS 4 : HMonths_L = "Rabi al-thani"
CAS 5 : HMonths_L = "Jumada al-oula"
CAS 6 : HMonths_L = "Jumada al-akhira"
CAS 7 : HMonths_L = "Rajab"
CAS 8 : HMonths_L = "Shaaban"
CAS 9 : HMonths_L = "Ramadan"
CAS 10 : HMonths_L = "Shawwal"
CAS 11 : HMonths_L = "Dhu al-Qidah"
CAS 12 : HMonths_L = "Dhu al-Hijjah"
```

FIN

DateHejri est une chaîne = d+" "+ HMonths\_L +" " + y

DateHejri2 est une chaîne = d+" / "+ m +" / " + y

Info(DateHejri,DateHejri2)//vous pouvez utiliser un renvoyer DateHejri ou DateHejri2

Sommaire > Concepts, WLangage et environnement > Divers > Divers

### Envoyer un mail avec l'emailer par défaut

**Auteurs : Emmanuel Lecoester ,**

```
sCourriel est une chaîne = "adressesmel"  
sMonsujet est une chaîne = "Sujet"  
sMontexte est une chaîne = "Contenu"  
  
LanceAppliAssociée("mailto:"+sCourriel+"?subject="+sMonsujet+"_etcommercial_body="+sMontexte)  
  
sMontexte=remplace(sMontexte,RC,"%0d%0a") //Permet d'envoyer un texte multiligne.
```

### Comment savoir si une adresse email est valide ?

**Auteurs : Emmanuel Lecoester ,**

**En WinDev 14**

```
EmailVérifieAdresse()
```

Elle permet : - Une vérification syntaxique ou - Une vérification syntaxique et du TLD (.fr, .com, etc...) ou - Une vérification syntaxique et du TLD (.fr, .com, etc...) et test de l'adresse (Il faut avoir accès à internet : interrogation SMTP)

### Que conserver d'un projet : organisation des répertoires d'un projet

**Auteurs : Emmanuel Lecoester ,**

**Il y a toute l'information détaillée sur les fichiers WD, leur utilité, ceux qui faut conserver et ceux que l'on peut supprimer dans l'aide Aide -- Index -- Projet -- Organisation des répertoires d'un projet**

### Exécuter une macro dans Excel

**Auteurs : Emmanuel Lecoester ,**

```
gclExcel est un objet cExcel()  
gclExcel:Ouvrir(sNomClasseur)  
gclExcel:m_xExcel>>RUN(sMacro)
```

**cExcel est la classe standard fournie avec Windev (voir "C:\WinDev 14\Exemples\Exemples didactiques\WD Pilotage de Excel"**

**sNomClasseur est un chaîne contenant le nom complet du document à ouvrir**

**sMcro est une chaîne contenant le nom de la macro (incluse dans le document)**